/*
 *  Jajuk
 *  Copyright (C) 2003-2008 The Jajuk Team
 *
 *  This program is free software; you can redistribute it and/or
 *  modify it under the terms of the GNU General Public License
 *  as published by the Free Software Foundation; either version 2
 *  of the License, or any later version.
 *
 *  This program is distributed in the hope that it will be useful,
 *  but WITHOUT ANY WARRANTY; without even the implied warranty of
 *  MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
 *  GNU General Public License for more details.
 *
 *  You should have received a copy of the GNU General Public License
 *  along with this program; if not, write to the Free Software
 *  Foundation, Inc., 59 Temple Place - Suite 330, Boston, MA  02111-1307, USA.
 *  $Revision: 3132 $
 */
package aspecto;

import ext.services.network.NetworkUtils;

import java.io.File;
import java.io.IOException;
import java.net.HttpURLConnection;
import java.net.URL;

import org.jajuk.util.*;

public aspect ConnectionAspect {
  
  public static HttpURLConnection DownloadManager.connection;
  
  pointcut connectionDownload(URL url) :
    execution(void DownloadManager.download(URL,..)) && args(url,..) ||
    execution(File DownloadManager.downloadCover(URL,..)) && args(url,..);
  
  declare soft : IOException : withincode(void DownloadManager.download(..)) ||
                               withincode(File DownloadManager.downloadCover(..));
  
  before(URL url) : connectionDownload(url){
      try {
        DownloadManager.connection = NetworkUtils.getConnection(url, DownloadManager.getProxy());
      } catch (IOException e) {
        throw new org.aspectj.lang.SoftException(e);
      }
  }
  
  after(URL url) : connectionDownload(url){
      DownloadManager.connection.disconnect();
  }
}
